
	Other comments:
	


1. Corrections required
This section describes the TTCN changes required to ‘iwd-TTCN-B2013-03_D14wk10’ release.
Change 1 – f_IMS_PTC_Deregistration
	Function name
	 f_IMS_PTC_Deregistration

	Reason for change
	User initiated de-registration requests are described in 24.229, section 5.1.1.6. There is no mentioning of mmtel feature-param in the Contact header (while it is mentioned for initial and re-registration). This is reflected in 34.229-1, C.30 by not listing the feature params in the Specific message contents section. Unluckily, 34.229-1, A.1.1 seems to allow the interpretation that A4 will apply also in case of de-registration (as A4 is a time-invariant condition). 
Note: prose CR R5-142582 is being submitted and planned r1 of that CR will add a clarifying Note to C.30.

	Summary of change
	We need to provide the code checking the Contact header with the information if this is a de-registration request or not. To this end, a Boolean argument is added to function f_IMS_A_1_1_MessageHeader_CommonChecks (and further forwarded, see below changes)

	TTCN module
	 Common/IMS/IMS_Procedures_Registration

	MCC160 Comment
	


Before change

    function f_IMS_PTC_Deregistration(template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit) runs on IMS_PTC

  { /* @sic R5-140902: no delay needed anymore to wait for TCP close sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var MessageHeader v_MessageHeader;

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var InternetProtocol_Type v_Protocol;

    var charstring v_Expires := "0";

    var boolean v_Match := false;

    v_IMS_DATA_REQ := fl_IMS_ReceiveOrGetRegisterRequest(p_IMS_DATA_REQ);

    v_MessageHeader := v_IMS_DATA_REQ.Request.Register.msgHeader;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol;

    v_Match := match(v_MessageHeader, f_IMS_RegisterRequest_MessageHeaderRX_Derregister());

    if (not v_Match) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid REGISTER Request for de-registration");

    }

    else {

      f_IMS_A_1_1_MessageHeader_CommonChecks(v_MessageHeader, v_Protocol, -, v_UE_Address, v_Expires);   // check expires: to be 0 and present either in Expires or in contact or in both (-> 8.3.4/8.13.4)

    }

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo),

                                     cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_IMS_DATA_REQ.Request.Register, -, v_Expires))));  // @sic R5s130274: v_Expires sic@

  }
After change

  function f_IMS_PTC_Deregistration(template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit) runs on IMS_PTC

  { /* @sic R5-140902: no delay needed anymore to wait for TCP close sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var MessageHeader v_MessageHeader;

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var InternetProtocol_Type v_Protocol;

    var charstring v_Expires := "0";

    var boolean v_Match := false;

    v_IMS_DATA_REQ := fl_IMS_ReceiveOrGetRegisterRequest(p_IMS_DATA_REQ);

    v_MessageHeader := v_IMS_DATA_REQ.Request.Register.msgHeader;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol;

    v_Match := match(v_MessageHeader, f_IMS_RegisterRequest_MessageHeaderRX_Derregister());

    if (not v_Match) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid REGISTER Request for de-registration");

    }

    else {

      f_IMS_A_1_1_MessageHeader_CommonChecks(v_MessageHeader, v_Protocol, -, v_UE_Address, v_Expires, true);  // check expires: to be 0 and present either in Expires or in contact or in both (-> 8.3.4/8.13.4)

    }

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo),

                                     cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_IMS_DATA_REQ.Request.Register, -, v_Expires))));  // @sic R5s130274: v_Expires sic@

  }   
Change 2 – f_IMS_A_1_1_MessageHeader_CommonChecks
	Function name
	 f_IMS_A_1_1_MessageHeader_CommonChecks

	Reason for change
	 Forward Boolean indicator

	Summary of change
	 Add the Boolean indicator in function definition of f_IMS_A_1_1_MessageHeader_CommonChecks and put it into embedded call to fl_IMS_Register_ContactAddress_CheckParams where it will finally be needed.

	TTCN module
	 Common/IMS/IMS_Procedures_Registration

	MCC160 Comment
	


Before change

  function f_IMS_A_1_1_MessageHeader_CommonChecks(MessageHeader p_MessageHeader,

                                                  InternetProtocol_Type p_Protocol,

                                                  boolean p_CondA7 := false,

                                                  template (omit) IP_AddrInfo_Type p_UE_Address := omit,

                                                  template (omit) charstring p_ExplicitExpiry := omit,

                                                  boolean p_CheckAuthorization := true) runs on IMS_PTC

  {

    var ContactAddress v_ContactAddress;

    f_IMS_MessageHeader_CheckVia(p_MessageHeader, p_Protocol);  // @sic R5s120727 change 7, 8 sic@

    f_IMS_MessageHeader_CheckContactAddr(p_MessageHeader);      // @sic R5s120727 change 7, 8 sic@

    // Check 'To'

    f_IMS_MessageHeader_CheckTo(p_MessageHeader, noToTag);

    // Check 'ContactAddress'

    if (ispresent(p_MessageHeader.contact.contactBody.contactAddresses[0])) {     /* NOTE: is not be present when 'contact' is wildcard e.g. in case of de-registration

                                                                                   *       but in this case the respective receive template needs to allow 'cr_ContactWildcard' */

      v_ContactAddress := p_MessageHeader.contact.contactBody.contactAddresses[0];

      fl_IMS_Register_ContactAddress_CheckParams(v_ContactAddress, p_CondA7);

      if (isvalue(p_UE_Address)) {

        fl_IMS_ContactAddress_CheckHostname(v_ContactAddress, valueof(p_UE_Address));

      }

    }

    // Check 'SecurityClient'

    if (ispresent(p_MessageHeader.securityClient)) {                               /* 'SecurityClient' is only included for A1, A2 but not for A3 */

      fl_IMS_SecurityClient_Check(p_MessageHeader.securityClient);

    }

    // Check 'Authorization'

    if (p_CheckAuthorization and ispresent(p_MessageHeader.authorization)) {       /* 'Authorization' is only included for A2 but optional for A1, A3 */

      fl_IMS_Authorization_Check(p_MessageHeader.authorization);

    }

    fl_IMS_MessageHeader_CheckExpires(p_MessageHeader, p_ExplicitExpiry);

  }
After change

   function f_IMS_A_1_1_MessageHeader_CommonChecks(MessageHeader p_MessageHeader,

                                                  InternetProtocol_Type p_Protocol,

                                                  boolean p_CondA7 := false,

                                                  template (omit) IP_AddrInfo_Type p_UE_Address := omit,

                                                  template (omit) charstring p_ExplicitExpiry := omit,

                                                  boolean p_CheckAuthorization := true,

                                                  boolean p_isDeregistration := false  

                                                  ) runs on IMS_PTC

  {

    var ContactAddress v_ContactAddress;

    f_IMS_MessageHeader_CheckVia(p_MessageHeader, p_Protocol);  // @sic R5s120727 change 7, 8 sic@

    f_IMS_MessageHeader_CheckContactAddr(p_MessageHeader);      // @sic R5s120727 change 7, 8 sic@

    // Check 'To'

    f_IMS_MessageHeader_CheckTo(p_MessageHeader, noToTag);

    // Check 'ContactAddress'

    if (ispresent(p_MessageHeader.contact.contactBody.contactAddresses[0])) {     /* NOTE: is not be present when 'contact' is wildcard e.g. in case of de-registration

                                                                                   *       but in this case the respective receive template needs to allow 'cr_ContactWildcard' */

      v_ContactAddress := p_MessageHeader.contact.contactBody.contactAddresses[0];

      fl_IMS_Register_ContactAddress_CheckParams(v_ContactAddress, p_CondA7, -, -, -, -, p_isDeregistration);  

      if (isvalue(p_UE_Address)) {

        fl_IMS_ContactAddress_CheckHostname(v_ContactAddress, valueof(p_UE_Address));

      }

    }

    // Check 'SecurityClient'

    if (ispresent(p_MessageHeader.securityClient)) {                               /* 'SecurityClient' is only included for A1, A2 but not for A3 */

      fl_IMS_SecurityClient_Check(p_MessageHeader.securityClient);

    }

    // Check 'Authorization'

    if (p_CheckAuthorization and ispresent(p_MessageHeader.authorization)) {       /* 'Authorization' is only included for A2 but optional for A1, A3 */

      fl_IMS_Authorization_Check(p_MessageHeader.authorization);

    }

    fl_IMS_MessageHeader_CheckExpires(p_MessageHeader, p_ExplicitExpiry);

  }
Change 3 – fl_IMS_Register_ContactAddress_CheckParams
	Function name
	 fl_IMS_Register_ContactAddress_CheckParams

	Reason for change
	 Do not request the mmtel feature param in case of user initiated de-registration 

	Summary of change
	Take received indicator. In case it specifies a de-registration, do not check for the mmtel feature param

	TTCN module
	 Common/IMS/IMS_Procedures_Registration

	MCC160 Comment
	


Before change

    function fl_IMS_Register_ContactAddress_CheckParams(ContactAddress p_ContactAddress,

                                                      boolean p_CondA7 := false,

                                                      boolean p_IMS_MTSI := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                                      boolean p_IMS_GRUUsInSIP := pc_IMS_GRUUsInSIP,

                                                      boolean p_IMS_SMSoverIMS := tsc_IMS_SMSoverIMS,

                                                      boolean p_Video := tsc_IMS_VideoFeature) runs on IMS_PTC

  { /* NOTE: 'Expires' can only be checked in context with the 'Expires' header field */

    var SemicolonParam_List v_ContactParams := {};

    var SemicolonParam_List v_UrlParams := {};

    var SipUrl v_SipUrl;

    if (ispresent(p_ContactAddress.contactParams)) {

      v_ContactParams := p_ContactAddress.contactParams;

    }

    if (p_CondA7) {

      v_SipUrl := f_Addr_Union_GetSipUrl(p_ContactAddress.addressField);

      if (ispresent(v_SipUrl.urlParameters)) {

        v_UrlParams := v_SipUrl.urlParameters;

      }

      if (not match(v_UrlParams, cr_SemicolonParam_List_OneSpecificParam("sos"))) {    // @sic R5s130495 change 10 sic@

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Missing sos URI parameter for 'ContactAddress'");

      }

    }

    // Rel-7 specific, @sic R5-073440 sic@

    if (p_IMS_MTSI) {                                                             // check 'feature-param'

      if (not match(v_ContactParams, cr_FeatureParamInContactParams(tsc_IMS_AcceptContactValue_Quoted))) { // @sic R5s130109, R5s130738: string needs to be quoted sic@

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'Contact Feature Params'");

      }

    }

    if (p_IMS_GRUUsInSIP) {                                                       // check 'c-p-instance'

      //!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

      // @sic R5-134659 sic@

      //!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

      if (not match(v_ContactParams, cr_CPInstanceInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }

    if (p_IMS_SMSoverIMS) {                                                       // check 'feature-param'

      if (not match(v_ContactParams, cr_SMSParamInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid SMS params in 'ContactAddress'");

      }

    }

    if (p_Video) {                                                       // check 'feature-param'

      if (not match(v_ContactParams, cr_VideoInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid video params in 'ContactAddress'");

      }

    }

  }
After change

  function fl_IMS_Register_ContactAddress_CheckParams(ContactAddress p_ContactAddress,

                                                      boolean p_CondA7 := false,

                                                      boolean p_IMS_MTSI := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                                      boolean p_IMS_GRUUsInSIP := pc_IMS_GRUUsInSIP,

                                                      boolean p_IMS_SMSoverIMS := tsc_IMS_SMSoverIMS,

                                                      boolean p_Video := tsc_IMS_VideoFeature,

                                                      boolean p_isDeregistration := false) 

                                                      runs on IMS_PTC

  { /* NOTE: 'Expires' can only be checked in context with the 'Expires' header field */

    var SemicolonParam_List v_ContactParams := {};

    var SemicolonParam_List v_UrlParams := {};

    var SipUrl v_SipUrl;

    if (ispresent(p_ContactAddress.contactParams)) {

      v_ContactParams := p_ContactAddress.contactParams;

    }

    if (p_CondA7) {

      v_SipUrl := f_Addr_Union_GetSipUrl(p_ContactAddress.addressField);

      if (ispresent(v_SipUrl.urlParameters)) {

        v_UrlParams := v_SipUrl.urlParameters;

      }

      if (not match(v_UrlParams, cr_SemicolonParam_List_OneSpecificParam("sos"))) {    // @sic R5s130495 change 10 sic@

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Missing sos URI parameter for 'ContactAddress'");

      }

    }

    // Rel-7 specific, @sic R5-073440 sic@
    if (p_IMS_MTSI and not p_isDeregistration) {    // check 'feature-param'

      if (not match(v_ContactParams, cr_FeatureParamInContactParams(tsc_IMS_AcceptContactValue_Quoted))) { // @sic R5s130109, R5s130738: string needs to be quoted sic@

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'Contact Feature Params'");

      }

    }

    if (p_IMS_GRUUsInSIP) {                                                       // check 'c-p-instance'

      //!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

      // @sic R5-134659 sic@

      //!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

      if (not match(v_ContactParams, cr_CPInstanceInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }

    if (p_IMS_SMSoverIMS) {                                                       // check 'feature-param'

      if (not match(v_ContactParams, cr_SMSParamInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid SMS params in 'ContactAddress'");

      }

    }

    if (p_Video) {                                                       // check 'feature-param'

      if (not match(v_ContactParams, cr_VideoInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid video params in 'ContactAddress'");

      }

    }

  }  
